The purpose of this worksheet is to study a differential time-delay linear system usually encountered in the process control literature and studied in N. Petit, Y. Creff, P. Rouchon, "δ-freeness of a class of linear systems", Proceedings of the 4th European Control Conference, Brussels (Belgium), 01-04/07/97.

The system is defined by means of the following transfer matrix:

```
> T := evalm([[K11*exp(-delta11*s)/(1+tau11*s), K12*exp(-delta12*s)/(1+tau12*s)],
> [K21*exp(-delta21*s)/(1+tau21*s), K22*exp(-delta22*s)/(1+tau22*s)]]);
```

$$T := \begin{bmatrix} \frac{K11\, e^{(-\delta 11\, s)}}{1+\tau 11\, s} & \frac{K12\, e^{(-\delta 12\, s)}}{1+\tau 12\, s} \\ \frac{K21\, e^{(-\delta 21\, s)}}{1+\tau 21\, s} & \frac{K22\, e^{(-\delta 22\, s)}}{1+\tau 22\, s} \end{bmatrix}$$

Let us firstly define the following algebra *Alg* of differential time-delay operators.

```
> Alg := DefineOreAlgebra(diff=[D,t], dual_shift=[Delta11,s1],
> dual_shift=[Delta12,s2], dual_shift=[Delta21,s3],
> dual_shift=[Delta22,s4], polynom=[t,s1,s2,s3,s4],
> shift_action=[Delta11,t,delta11], shift_action=[Delta12,t,delta12],
> shift_action=[Delta21,t,delta21], shift_action=[Delta22,t,delta22],
> comm=[K11,K12,K21,K22,tau11,tau12,tau21,tau22]):
```

Introducing the new variables $x[i,j]$, $i,j=1,2$, satisfying the equations

$$(1+\tau ij\, D)\, x[i,j](t) = Kij\, u_i(t-\delta ij), \quad i,j=1,2,$$

where D denotes the time-derivative and u_i, $i=1$, 2, the inputs of the system, we obtain a description of the transfer matrix T by means of the following matrix of differential time-delay operators (Δij denotes the time-delay operator of length δij):

```
> R := evalm([[1+tau11*D,0,0,0,-Delta11,0,0,0], [0,1+tau12*D,0,0,0,-Delta12,0,0],
> [0,0,1+tau21*D,0,-Delta21,0,0,0],  [0,0,0,1+tau22*D,0,-Delta22,0,0],
> [K11,K12,0,0,0,0,-1,0], [0,0,K21,K22,0,0,0,-1]]);
```

$$R := \begin{bmatrix} 1+\tau 11\,\mathrm{D} & 0 & 0 & 0 & -\Delta 11 & 0 & 0 & 0 \\ 0 & 1+\tau 12\,\mathrm{D} & 0 & 0 & 0 & -\Delta 12 & 0 & 0 \\ 0 & 0 & 1+\tau 21\,\mathrm{D} & 0 & -\Delta 21 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1+\tau 22\,\mathrm{D} & 0 & -\Delta 22 & 0 & 0 \\ K11 & K12 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & K21 & K22 & 0 & 0 & 0 & -1 \end{bmatrix}$$

Hence, we now consider the following differential time-delay system, where we denote by $z_1 = x[1,1]$, $z_2 = x[1,2]$, $z_3 = x[2,1]$, $z_4 = x[2,2]$, and y_i, $i=1$, 2, denote the outputs of the system:

```
> ApplyMatrix(R, [seq(z[i](t), i=1..4), seq(u[i](t), i=1..2),
> seq(y[i](t), i=1..2)], Alg)=evalm([seq([0],i=1..6)]);
```

$$\begin{bmatrix} z_1(t)+\tau 11\,\mathrm{D}(z_1)(t)-u_1(t-\delta 11) \\ z_2(t)+\tau 12\,\mathrm{D}(z_2)(t)-u_2(t-\delta 12) \\ z_3(t)+\tau 21\,\mathrm{D}(z_3)(t)-u_1(t-\delta 21) \\ z_4(t)+\tau 22\,\mathrm{D}(z_4)(t)-u_2(t-\delta 22) \\ K11\, z_1(t)+K12\, z_2(t)-y_1(t) \\ K21\, z_3(t)+K22\, z_4(t)-y_2(t) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

Let us study the algebraic properties of the previous system. For the sake of simplicity, as the inputs y_i, $i=1$, 2, are trivial elements of the *Alg*-module associated with the following matrix

```
> R2 := evalm([[1+tau11*D,0,0,0,-Delta11,0], [0,1+tau12*D,0,0,0,-Delta12],
> [0,0,1+tau21*D,0,-Delta21,0], [0,0,0,1+tau22*D,0,-Delta22]]);
```

$$
R2 := \begin{bmatrix} 1+\tau 11\,\mathrm{D} & 0 & 0 & 0 & -\Delta 11 & 0 \\ 0 & 1+\tau 12\,\mathrm{D} & 0 & 0 & 0 & -\Delta 12 \\ 0 & 0 & 1+\tau 21\,\mathrm{D} & 0 & -\Delta 21 & 0 \\ 0 & 0 & 0 & 1+\tau 22\,\mathrm{D} & 0 & -\Delta 22 \end{bmatrix}
$$

we may just consider the *Alg*-module associated with *R2*. In other words, the *Alg*-module associated with *R2* is the same as the *Alg*-module associated with *R*. Therefore, we only keep the equations describing the dynamics of the first system, namely:

```
> ApplyMatrix(R2, [seq(z[i](t), i=1..4), seq(u[i](t), i=1..2)], Alg)=
> evalm([seq([0],i=1..4)]);
```

$$
\begin{bmatrix} z_1(t)+\tau 11\,\mathrm{D}(z_1)(t)-u_1(t-\delta 11) \\ z_2(t)+\tau 12\,\mathrm{D}(z_2)(t)-u_2(t-\delta 12) \\ z_3(t)+\tau 21\,\mathrm{D}(z_3)(t)-u_1(t-\delta 21) \\ z_4(t)+\tau 22\,\mathrm{D}(z_4)(t)-u_2(t-\delta 22) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}
$$

Let us define the adjoint *R_adj* of *R*.

```
> R_adj := Involution(R2, Alg):
```

Then, we know that the module properties of the *Alg*-module associated with *R2* correspond to the vanishing of some extension modules of the *Alg*-module associated with *R_adj* with values in *Alg*. Let us compute the first extension module of this *Alg*-module.

```
> Ext1 := Exti(R_adj, Alg, 1);
```

$$
Ext1 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1+\tau 11\,\mathrm{D} & 0 & 0 & 0 & -\Delta 11 & 0 \\ 0 & 1+\tau 12\,\mathrm{D} & 0 & 0 & 0 & -\Delta 12 \\ 0 & 0 & 1+\tau 21\,\mathrm{D} & 0 & -\Delta 21 & 0 \\ 0 & 0 & 0 & 1+\tau 22\,\mathrm{D} & 0 & -\Delta 22 \end{bmatrix}, \right.
$$

$$
\left. \begin{bmatrix} \tau 21\,\Delta 11\,\mathrm{D}+\Delta 11 & 0 \\ 0 & \tau 22\,\Delta 12\,\mathrm{D}+\Delta 12 \\ \tau 11\,\mathrm{D}\,\Delta 21+\Delta 21 & 0 \\ 0 & \tau 12\,\mathrm{D}\,\Delta 22+\Delta 22 \\ \tau 21\,\mathrm{D}+\tau 21\,\mathrm{D}^2\,\tau 11+1+\tau 11\,\mathrm{D} & 0 \\ 0 & \tau 22\,\mathrm{D}+\tau 22\,\mathrm{D}^2\,\tau 12+1+\tau 12\,\mathrm{D} \end{bmatrix} \right]
$$

As the first matrix *Ext1*[1] is the identity matrix, we obtain that the *Alg*-module associated with *R2* is torsion-free. Therefore, we find that the system defined by *R2* is controllable and parametrizable.

A parametrization of the system defined by *R2* is then given by *Ext1*[3]. Equivalently, such a parametrization can be directly computed using the command *Parametrization*:

```
> evalm([seq([z[i](t)], i=1..4), seq([u[i](t)], i=1..2)])=
> Parametrization(R2, Alg);
```

$$
\begin{bmatrix} z_1(t) \\ z_2(t) \\ z_3(t) \\ z_4(t) \\ u_1(t) \\ u_2(t) \end{bmatrix} = \begin{bmatrix} \tau 21\,\mathrm{D}(\xi_1)(t-\delta 11)+\xi_1(t-\delta 11) \\ \tau 22\,\mathrm{D}(\xi_2)(t-\delta 12)+\xi_2(t-\delta 12) \\ \tau 11\,\mathrm{D}(\xi_1)(t-\delta 21)+\xi_1(t-\delta 21) \\ \tau 12\,\mathrm{D}(\xi_2)(t-\delta 22)+\xi_2(t-\delta 22) \\ \tau 21\,\mathrm{D}(\xi_1)(t)+\tau 21\,\tau 11\,(\mathrm{D}^{(2)})(\xi_1)(t)+\xi_1(t)+\tau 11\,\mathrm{D}(\xi_1)(t) \\ \tau 22\,\mathrm{D}(\xi_2)(t)+\tau 22\,\tau 12\,(\mathrm{D}^{(2)})(\xi_2)(t)+\xi_2(t)+\tau 12\,\mathrm{D}(\xi_2)(t) \end{bmatrix}
$$

If the parametrization *Ext1*[3] admits a left-inverse, then the *Alg*-module associated with *R2* is free.

```
> LeftInverse(Ext1[3], Alg);
```

$$[]$$

As no left-inverse of *Ext1*[3] exists, we cannot conclude about the freeness of the *Alg*-module associated with *R2*. Let us check whether or not such a module is reflexive by computing the second extension module.

```
> Ext2 := Exti(R_adj, Alg, 2);
```

$$Ext2 := \left[\left[\begin{array}{cc} \Delta 21\,\Delta 11 & 0 \\ \tau 11\,\mathrm{D}\,\Delta 21 + \Delta 21 & 0 \\ \tau 21\,\Delta 11\,\mathrm{D} + \Delta 11 & 0 \\ \tau 21\,\mathrm{D} + \tau 21\,\mathrm{D}^2\,\tau 11 + 1 + \tau 11\,\mathrm{D} & 0 \\ 0 & \Delta 22\,\Delta 12 \\ 0 & \tau 12\,\mathrm{D}\,\Delta 22 + \Delta 22 \\ 0 & \tau 22\,\Delta 12\,\mathrm{D} + \Delta 12 \\ 0 & \tau 22\,\mathrm{D} + \tau 22\,\mathrm{D}^2\,\tau 12 + 1 + \tau 12\,\mathrm{D} \end{array}\right], \left[\begin{array}{cc} 1 & 0 \\ 0 & 1 \end{array}\right], \mathrm{SURJ}(2)\right]$$

As the first matrix *Ext2*[1] is not an identity matrix, we obtain that the second extension module is not trivial, and thus, the *Alg*-module associated with *R2* is torsion-free but not reflexive. In particular, we deduce that the *Alg*-module is not free, and thus, the system defined by *R2* is not flat. However, as the torsion-free degree of the system *R2* is 1, we know that it is π-free. Let us compute such a polynomial π.

```
> pi := PiPolynomial(R2, Alg, [Delta11,Delta12,Delta21,Delta22]);
```

$$\pi := [\Delta 21\,\Delta 11\,\Delta 22\,\Delta 12]$$

Hence, if we localize the algebra *Alg* with respect to the polynomial π in order to obtain the algebra *Alg2*, then the new module associated with *R2* over *Alg2* is free. Equivalently, if we can invert the time-delay operator π, i.e., use the time-advance operator π^{-1}, we then obtain a basis of the *Alg2*-module associated with *R2* as follows:

```
> S := LocalLeftInverse(Ext1[3], pi, Alg);
```

$$S := \left[\begin{array}{cccccc} \dfrac{\tau 11}{\Delta 11\,(\tau 11 - \tau 21)} & 0 & -\dfrac{\tau 21}{\Delta 21\,(\tau 11 - \tau 21)} & 0 & 0 & 0 \\ 0 & \dfrac{\tau 12}{\Delta 12\,(-\tau 22 + \tau 12)} & 0 & -\dfrac{\tau 22}{\Delta 22\,(-\tau 22 + \tau 12)} & 0 & 0 \end{array}\right]$$

We easily check that S is a left-inverse of the parametrization *Ext1*[3] over the algebra *Alg2*:

```
> simplify(evalm(S &* Ext1[3]));
```

$$\left[\begin{array}{cc} 1 & 0 \\ 0 & 1 \end{array}\right]$$

Then, a basis (ξ_1, ξ_2) of this new module is defined by:

```
> evalm([seq([xi[i](t)],i=1..2)])=ApplyMatrix(S,[seq(z[i](t), i=1..4),
> seq(u[i](t), i=1..2)], Alg);
```

$$\left[\begin{array}{c}\xi_1(t)\\ \xi_2(t)\end{array}\right] = \left[\begin{array}{c}\dfrac{\tau 11\, z_1(t+\delta 11) - \tau 21\, z_3(t+\delta 21)}{\tau 11 - \tau 21}\\ \dfrac{\tau 12\, z_2(t+\delta 12) - \tau 22\, z_4(t+\delta 22)}{-\tau 22 + \tau 12}\end{array}\right]$$

In particular, we check that such a basis exists if $\tau 11 \neq \tau 21$ and $\tau 12 \neq \tau 22$.

Now, we come back to the outputs y_i, $i = 1, 2$, of the system. If we define the following matrix

```
> K := evalm([[K11, K12, 0, 0], [0, 0, K21, K22]]);
```

$$K := \left[\begin{array}{cccc} K11 & K12 & 0 & 0\\ 0 & 0 & K21 & K22\end{array}\right]$$

then we obtain that the outputs y_i, $i = 1, 2$, are defined in terms of the variables z_k as follows:

```
> evalm([seq([y[i](t)],i=1..2)])=ApplyMatrix(K, [seq(z[i](t),i=1..4)], Alg);
```

$$\left[\begin{array}{c} y_1(t)\\ y_2(t)\end{array}\right] = \left[\begin{array}{c} K11\, z_1(t) + K12\, z_2(t)\\ K21\, z_3(t) + K22\, z_4(t)\end{array}\right]$$

Substituting $(z_1, z_2, z_3, z_4, u_1, u_2)^T = Ext1[3]\,(\xi_1, \xi_2)^T$ into the previous relations, we finally obtain the expressions for y_i, $i = 1, 2$, and u_j, $j = 1, 2$, in terms of the elements of the basis ξ_1 and ξ_2:

```
> L := linalg[submatrix](Ext1[3], 1..4, 1..2):  evalm([seq([y[i](t)],i=1..2)])=
> ApplyMatrix(Mult(K, L, Alg),[xi[1](t),xi[2](t)], Alg);
```

$$\left[\begin{array}{c} y_1(t)\\ y_2(t)\end{array}\right] = \left[\begin{array}{c} K11\, \xi_1(t-\delta 11) + K11\, \tau 21\, \mathrm{D}(\xi_1)(t-\delta 11) + K12\, \xi_2(t-\delta 12) + K12\, \tau 22\, \mathrm{D}(\xi_2)(t-\delta 12)\\ K21\, \xi_1(t-\delta 21) + K21\, \tau 11\, \mathrm{D}(\xi_1)(t-\delta 21) + K22\, \xi_2(t-\delta 22) + K22\, \tau 12\, \mathrm{D}(\xi_2)(t-\delta 22)\end{array}\right]$$

```
> evalm([seq([u[i](t)],i=1..2)])=ApplyMatrix(linalg[submatrix](Ext1[3],
> 5..6, 1..2),[xi[1](t),xi[2](t)], Alg);
```

$$\left[\begin{array}{c} u_1(t)\\ u_2(t)\end{array}\right] = \left[\begin{array}{c} \tau 21\, \mathrm{D}(\xi_1)(t) + \tau 21\, \tau 11\, (\mathrm{D}^{(2)})(\xi_1)(t) + \xi_1(t) + \tau 11\, \mathrm{D}(\xi_1)(t)\\ \tau 22\, \mathrm{D}(\xi_2)(t) + \tau 22\, \tau 12\, (\mathrm{D}^{(2)})(\xi_2)(t) + \xi_2(t) + \tau 12\, \mathrm{D}(\xi_2)(t)\end{array}\right]$$

We find again the parametrization of the system given in N. Petit, Y. Creff, P. Rouchon, "δ-freeness of a class of linear systems", Proceedings of the 4th European Control Conference, Brussels (Belgium), 01-04/07/97.

Now, we can wonder if we can express the elements of the basis ξ_1 and ξ_2 only in terms of the inputs u_i, $i = 1, 2$, and the outputs y_j, $j = 1, 2$. From the previous two systems, we have

$$(y_1, y_2, u_1, u_2)^T = Q\,(\xi_1, \xi_2)^T,$$

where the matrix Q is defined by:

```
> Q := linalg[stackmatrix](Mult(K, L, Alg), linalg[submatrix](Ext1[3], 5..6, 1..2));
```

$$Q := \left[\begin{array}{cc} K11\, \Delta 11\, (1 + \tau 21\, \mathrm{D}) & K12\, \Delta 12\, (1 + \tau 22\, \mathrm{D})\\ K21\, \Delta 21\, (1 + \tau 11\, \mathrm{D}) & K22\, \Delta 22\, (1 + \tau 12\, \mathrm{D})\\ \tau 21\, \mathrm{D} + \tau 21\, \mathrm{D}^2\, \tau 11 + 1 + \tau 11\, \mathrm{D} & 0\\ 0 & \tau 22\, \mathrm{D} + \tau 22\, \mathrm{D}^2\, \tau 12 + 1 + \tau 12\, \mathrm{D}\end{array}\right]$$

Let us check whether or not the matrix Q is left-invertible over *Alg2*.

```
> T := LocalLeftInverse(Q, pi, Alg);
```

$$T := \left[\frac{(1+\tau 12\,\mathrm{D})\,\tau 11^2}{\Delta 11\,(-\tau 21\,\tau 11+\tau 21\,\tau 12-\tau 12\,\tau 11+\tau 11^2)\,K11},\right.$$
$$-\frac{(1+\tau 22\,\mathrm{D})\,\tau 21^2}{\Delta 21\,(-\tau 22\,\tau 11+\tau 21\,\tau 11+\tau 21\,\tau 22-\tau 21^2)\,K21},-($$
$$\tau 21\,\tau 12\,\tau 11-\tau 22\,\tau 12\,\tau 11+\tau 22\,\tau 12\,\tau 21-\tau 11\,\tau 22\,\tau 21)/(\tau 22\,\tau 12\,\tau 11$$
$$-\tau 22\,\tau 12\,\tau 21-\tau 22\,\tau 11^2+\tau 11\,\tau 22\,\tau 21-\tau 21\,\tau 12\,\tau 11+\tau 12\,\tau 21^2+\tau 21\,\tau 11^2$$
$$-\tau 11\,\tau 21^2),(\tau 22\,K21\,\Delta 21\,\tau 11^2\,K12\,\Delta 12-\tau 21\,K21\,\Delta 21\,\tau 11^2\,K12\,\Delta 12$$
$$+\Delta 11\,K11\,\tau 21^2\,\tau 11\,K22\,\Delta 22-\Delta 11\,\tau 12\,K11\,\tau 21^2\,K22\,\Delta 22)\Big/(\Delta 21\,\Delta 11\,K21$$
$$K11(\tau 22\,\tau 12\,\tau 11-\tau 22\,\tau 12\,\tau 21-\tau 22\,\tau 11^2+\tau 11\,\tau 22\,\tau 21-\tau 21\,\tau 12\,\tau 11$$
$$\left.+\tau 12\,\tau 21^2+\tau 21\,\tau 11^2-\tau 11\,\tau 21^2))\right]$$
$$\left[-\frac{(1+\tau 11\,\mathrm{D})\,\tau 12^2}{\Delta 12\,(\tau 22\,\tau 12-\tau 22\,\tau 11-\tau 12^2+\tau 12\,\tau 11)\,K12},\right.$$
$$\frac{(1+\tau 21\,\mathrm{D})\,\tau 22^2}{\Delta 22\,(\tau 21\,\tau 12-\tau 22\,\tau 12-\tau 21\,\tau 22+\tau 22^2)\,K22},(\Delta 12\,\tau 22^2\,K12\,\tau 12\,K21\,\Delta 21$$
$$-\Delta 12\,\tau 22^2\,\tau 11\,K12\,K21\,\Delta 21-K22\,\tau 12^2\,K11\,\Delta 22\,\Delta 11\,\tau 22$$
$$+K22\,\tau 12^2\,K11\,\Delta 22\,\Delta 11\,\tau 21)\Big/(\Delta 22\,\Delta 12\,K22\,K12(\tau 21\,\tau 12\,\tau 11-\tau 22\,\tau 12\,\tau 11$$
$$+\tau 22\,\tau 12^2-\tau 21\,\tau 12^2-\tau 11\,\tau 22\,\tau 21+\tau 22\,\tau 12\,\tau 21-\tau 22^2\,\tau 12+\tau 11\,\tau 22^2)),($$
$$\tau 21\,\tau 12\,\tau 11-\tau 22\,\tau 12\,\tau 11+\tau 22\,\tau 12\,\tau 21-\tau 11\,\tau 22\,\tau 21)/(\tau 21\,\tau 12\,\tau 11$$
$$-\tau 22\,\tau 12\,\tau 11+\tau 22\,\tau 12^2-\tau 21\,\tau 12^2-\tau 11\,\tau 22\,\tau 21+\tau 22\,\tau 12\,\tau 21-\tau 22^2\,\tau 12$$
$$\left.+\tau 11\,\tau 22^2)\right]$$

Therefore, we obtain that such a left-inverse of Q exists as we have:

```
> simplify(evalm(T &* Q));
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Hence, we can now express the elements of the basis ξ_1 and ξ_2 in terms of y_i, $i = 1, 2$, and u_j, $j = 1, 2$ as follows:

```
> X := ApplyMatrix(T, [seq([y[i](t)],i=1..2), seq([u[i](t)],i=1..2)], Alg):
> xi[1](t)=X[1,1];
```

$$\xi_1(t) = -(-\tau 11^2\,\tau 21\,y_1(t+\delta 11)\,K21-\tau 11^2\,\tau 21\,\tau 12\,\mathrm{D}(y_1)(t+\delta 11)\,K21$$
$$+\tau 21\,\tau 11^2\,K12\,u_2(t-\delta 12+\delta 11)\,K21+\tau 11^2\,y_1(t+\delta 11)\,K21\,\tau 22$$
$$+\tau 11^2\,\tau 12\,\mathrm{D}(y_1)(t+\delta 11)\,K21\,\tau 22-\tau 22\,\tau 11^2\,K12\,u_2(t-\delta 12+\delta 11)\,K21$$
$$+\tau 11\,\tau 22\,\tau 21^2\,\mathrm{D}(y_2)(t+\delta 21)\,K11-\tau 11\,\tau 21^2\,K22\,u_2(t-\delta 22+\delta 21)\,K11$$
$$+\tau 11\,\tau 21^2\,y_2(t+\delta 21)\,K11-\tau 11\,\tau 21\,u_1(t)\,K11\,K21\,\tau 22$$
$$+\tau 11\,\tau 21\,u_1(t)\,K11\,K21\,\tau 12-\tau 11\,u_1(t)\,K11\,K21\,\tau 22\,\tau 12$$
$$+\tau 12\,\tau 21^2\,K22\,u_2(t-\delta 22+\delta 21)\,K11-\tau 21^2\,\tau 22\,\mathrm{D}(y_2)(t+\delta 21)\,K11\,\tau 12$$
$$-\tau 21^2\,y_2(t+\delta 21)\,K11\,\tau 12+u_1(t)\,K11\,K21\,\tau 22\,\tau 21\,\tau 12)\Big/(K21\,K11($$
$$\tau 22\,\tau 12\,\tau 11-\tau 22\,\tau 12\,\tau 21-\tau 22\,\tau 11^2+\tau 11\,\tau 22\,\tau 21-\tau 21\,\tau 12\,\tau 11+\tau 12\,\tau 21^2$$
$$+\tau 21\,\tau 11^2-\tau 11\,\tau 21^2))$$

```
> xi[2](t)=X[2,1];
```

$$\xi_2(t) = (-\tau 12^2\,\tau 21\,y_1(t+\delta 12)\,\mathit{K22} + \tau 12^2\,\mathit{K11}\,\tau 21\,u_1(t-\delta 11+\delta 12)\,\mathit{K22}
- \tau 12^2\,\tau 21\,\tau 11\,\mathrm{D}(y_1)(t+\delta 12)\,\mathit{K22} + \tau 12^2\,y_1(t+\delta 12)\,\mathit{K22}\,\tau 22
- \tau 12^2\,\tau 22\,\mathit{K11}\,u_1(t-\delta 11+\delta 12)\,\mathit{K22} + \tau 12^2\,\tau 11\,\mathrm{D}(y_1)(t+\delta 12)\,\mathit{K22}\,\tau 22
+ \tau 12\,\tau 21\,\tau 11\,u_2(t)\,\mathit{K12}\,\mathit{K22} - \tau 12\,\tau 22^2\,\tau 21\,\mathrm{D}(y_2)(t+\delta 22)\,\mathit{K12}
+ \tau 12\,u_2(t)\,\mathit{K12}\,\mathit{K22}\,\tau 22\,\tau 21 + \mathit{K21}\,\tau 22^2\,\tau 12\,u_1(t-\delta 21+\delta 22)\,\mathit{K12}
- \tau 12\,\tau 22^2\,y_2(t+\delta 22)\,\mathit{K12} - \tau 12\,\tau 11\,u_2(t)\,\mathit{K12}\,\mathit{K22}\,\tau 22
- \tau 11\,u_2(t)\,\mathit{K12}\,\mathit{K22}\,\tau 22\,\tau 21 + \tau 21\,\tau 11\,\tau 22^2\,\mathrm{D}(y_2)(t+\delta 22)\,\mathit{K12}
- \mathit{K21}\,\tau 11\,\tau 22^2\,u_1(t-\delta 21+\delta 22)\,\mathit{K12} + \tau 11\,\tau 22^2\,y_2(t+\delta 22)\,\mathit{K12}) \Big/ (\mathit{K22}\,\mathit{K12}(
\tau 21\,\tau 12\,\tau 11 - \tau 22\,\tau 12\,\tau 11 + \tau 22\,\tau 12^2 - \tau 21\,\tau 12^2 - \tau 11\,\tau 22\,\tau 21 + \tau 22\,\tau 12\,\tau 21
- \tau 22^2\,\tau 12 + \tau 11\,\tau 22^2))$$

Finally, we note that the previous two expressions are well defined only if the two following denominators do not vanish:

```
> a := factor(denom(X[1,1]));
```

$$a := \mathit{K21}\,\mathit{K11}\,(-\tau 22 + \tau 21)\,(\tau 11 - \tau 21)\,(-\tau 12 + \tau 11)$$

```
> b := factor(denom(X[2,1]));
```

$$b := \mathit{K22}\,\mathit{K12}\,(-\tau 22 + \tau 21)\,(-\tau 22 + \tau 12)\,(-\tau 12 + \tau 11)$$

i.e., if the $\tau\ ij$'s are pairwise different.